Problem 2.2

Problem 2.2. Suppose you flip 20 fair coins.

(a) How many possible outcomes (microstates) are there?

(b) What is the probability of getting the sequence HTHHTTTHTHHHTHH-HHTHT (in exactly that order)?

(c) What is the probability of getting 12 heads and 8 tails (in any order)?

Consider flipping 20 fair coins. The Mathematica function Tuples[] will generate all permutations of the set {“H”, “T”}.

a. Microstates

In[∘]:=
```
microstates = Tuples[{"H", "T"}, 20];
```

In[∘]:=
```
TableForm[microstates⟦Range[10]⟧] (* Here are the first 10 elements *)
```

Out[∘]//TableForm=

H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H
H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H
H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	T
H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	T
H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	T	H
H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	T	H
H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	T	T
H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	T	T
H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	T	H	H
H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	H	T	H	H

In[∘]:=
```
nstates = Length[microstates] (* Note this is the same as 2^20 *)
```

Out[∘]=

1 048 576

macrostates (not part of this problem)

Count macrostates with the ‘Select’ function. For example, to find all states with 3 “H”, we simply Select[ ] all microstates that have 3 “H”s. Then the Length[ ] function can be used to count the length of that list, i.e. 4 microstates.

In[∘]:=
```
Select[microstates, Count[#, "H"] == 2 &];
```

In[∘]:=
```
Length[Select[microstates, Count[#, "H"] == 2 &]]
```

Out[∘]=

190

In[∘]:=
```
macrostates =
  Table[{n, multiplicity = Length[Select[microstates, Count[#, "H"] == n &]],
    N[multiplicity / nstates]}, {n, 0, 20}];
```

In[∘]:=
```
TableForm[macrostates,
 TableHeadings → {None, {"Number of Heads", "Multiplicity", "Probability"}}]
```

Out[∘]//TableForm=

Number of Heads	Multiplicity	Probability
0	1	9.53674×10^{-7}
1	20	0.0000190735
2	190	0.000181198
3	1140	0.00108719
4	4845	0.00462055
5	15 504	0.0147858
6	38 760	0.0369644
7	77 520	0.0739288
8	125 970	0.120134
9	167 960	0.160179
10	184 756	0.176197
11	167 960	0.160179
12	125 970	0.120134
13	77 520	0.0739288
14	38 760	0.0369644
15	15 504	0.0147858
16	4845	0.00462055
17	1140	0.00108719
18	190	0.000181198
19	20	0.0000190735
20	1	9.53674×10^{-7}

This is the same result we get from the combinatoric function (Binomial[n, m] in Mathematica).

In[ ]:=
```
TableForm[
 Table[{m, multiplicity = Binomial[20, m], N[multiplicity / 2^20]}, {m, 0, 20}],
 TableHeadings → {None, {"Number of Heads", "Multiplicity", "Probability"}}]
```

Out[ ]//TableForm=

Number of Heads	Multiplicity	Probability
0	1	9.53674×10^{-7}
1	20	0.0000190735
2	190	0.000181198
3	1140	0.00108719
4	4845	0.00462055
5	15 504	0.0147858
6	38 760	0.0369644
7	77 520	0.0739288
8	125 970	0.120134
9	167 960	0.160179
10	184 756	0.176197
11	167 960	0.160179
12	125 970	0.120134
13	77 520	0.0739288
14	38 760	0.0369644
15	15 504	0.0147858
16	4845	0.00462055
17	1140	0.00108719
18	190	0.000181198
19	20	0.0000190735
20	1	9.53674×10^{-7}

In[ ]:=
```
ListPlot[macrostates⟦All, {1, 3}⟧,
 AxesLabel → {"Number of Heads", "Probability"},
 LabelStyle → Larger, ImageSize → Scaled[0.8]]
```

Out[ ]=



b. Probability of getting a specific microstate

In[ ]:=
```
target = StringSplit["HTHHTTTHTHHHTHHHHTHT", ""]
```
Out[ ]=
```
{H, T, H, H, T, T, T, H, T, H, H, H, T, H, H, H, H, T, H, T}
```
In[ ]:=
```
Select[microstates, # == target &]
```
Out[ ]=
```
{{H, T, H, H, T, T, T, H, T, H, H, H, T, H, H, H, H, T, H, T}}
```
Since there is just one such state, the probability of getting it is

In[ ]:=
```
N[Length[%] / nstates]
```
Out[ ]=

9.53674×10^{-7}

c. Probability of getting a particular macrostate with 12 Heads

In[ ]:=
```
n12H = Length[Select[microstates, Count[#, "H"] == 12 &]]
```
Out[ ]=
```
125 970
```
In[ ]:=
```
p = N[n12H / nstates]
```
Out[ ]=
```
0.120134
```
Calculating the probability more directly using the Binomial coefficient:

In[ ]:=
```
N[Binomial[20, 12] / nstates]
```
Out[ ]=
```
0.120134
```